
[image: ]ISO/IEC JTC 1/SC 29/WG 03 N1054


ISO/IEC JTC 1/SC 29/WG 03
MPEG Systems 
Convenorship: KATS (Korea, Republic of)

Document type:	Output Document
Title:	Technology under Consideration on ISO/IEC 14496-34 Syntactic Description Language
Status:	Approved
Date of document:	2023-10-20
Source:	ISO/IEC JTC 1/SC 29/WG 03
No. of pages:	5 (with cover page)
Email of Convenor:	young.L @ samsung . com
Committee URL:	https://isotc.iso.org/livelink/livelink/open/jtc1sc29wg3


INTERNATIONAL ORGANIZATION FOR STANDARDIZATION
ORGANISATION INTERNATIONALE DE NORMALISATION
ISO/IEC JTC 1/SC 29/WG 03 MPEG SYSTEMS
ISO/IEC JTC 1/SC 29/WG 03 N1054
October 2023, Hannover, DE
	Title
	Technology under Consideration on ISO/IEC 14496-34 Syntactic Description Language

	Source
	WG 03, MPEG Systems

	Status
	Approved

	Serial Number
	23207




[bookmark: _Toc140841803]Introduction

This document collects technologies being under study for consideration in the development of the standard ISO/IEC 14496-34 Syntactic Description Language

General

This clause provides the SDL grammar implementing the rules of the present specification. The grammar is based on the parsing expression grammar (PEG) concept and following the syntax and set of rules defined by the Pegen software project [1].

Currently, the grammar file is hosted an developed at http://mpegx.int-evry.fr/software/MPEG/Systems/sdl/sdl-grammar.

The usage of PEG seems adequate for the validation of the SDL syntax, however there are still some questions whether the PEG grammar defined by the Pegen software project is appropriate for the developing the conformance of the SDL specification. Further study is encourage on other possible alternative for such grammar file.

Grammar file

[Editor’s note] The following is in work-in-progress and may not reflect all the rules described in the latest SDL specification text.


SDL grammar file
	start: file_input

file_input: (NEWLINE+ | line)* ENDMARKER

line: class_def | comment_cpp

#NOTE: 14496-1 forbids going back to a new line berfore {
# Rule C.1 and C.2
class_def: aligned? abstract? 'class' NAME+ parameter_list? parent_class? NEWLINE? '{' body? '}'

aligned: 'aligned' '(' NUMBER+ ')'

abstract: 'abstract'

parameter_list: '(' ','.parameter+ ')'

# NOTE: we allow arrays to be passed as parameter class, however 14496-1 is not clear on this
parameter: optional? type NAME array_length?

# NOTE: Not in 14496-1, not sure where this come from
optional: 'optional'

# NOTE: would allow "unsigned bit", to be improved
type: signed? data_type

signed: 'unsigned'

data_type:
  | 'bit'
  | 'int'
  | 'double'

array_length: '[' NUMBER* ']'

#NOTE: 14496-1 does not allow paramters after parent class name
parent_class: 'extends' NAME '(' ','.value+ ')'

body: stmt*

stmt:
  | elementary_data_type
  | non_parsable_variable
  | assignment_stmt
  | object_instantiation
  | increment_stmt
  | if_stmt
  | switch_stmt
  | for_stmt
  | do_stmt
  | while_stmt
  | comment_cpp

#TODO: See how to do any character up to newline
comment_cpp: '//' (NAME | 'floor' | 'class' | 'if' | 'else' | 'for' | 'extends' | NUMBER | '==' | '=' | '{' | ';' | ',' | '-' | '/' | ':' | '?'  )*

# Rule E.1 and A.1
elementary_data_type: template? aligned? const? type length NAME array_length? assigned_value? ';'

non_parsable_variable: template? const? type NAME array_length? assigned_value? ';'

#NOTE: Not in 14496-1 but used in 14496-12 
template: 'template'

const: 'const'

length: '(' (NUMBER | NAME) ')'

#NOTE: array initialisation with {val1, val2, ...} not in 14496-1
assigned_value: '=' (value | array_initialisation)

object_instantiation: NAME NAME ( '(' ','.value+ ')' )* array_length? ';'

variable_assignment: NAME assigned_value
assignment_stmt: variable_assignment ';'

#TODO: This rule should not allow whitepaces between name and '+'s
variable_incr: NAME '+' '+'

increment_stmt: variable_incr ';'

#NOTE: STRING literal e.g. 'uuid' is not allowed in 14496-1
#NOTE: NUMBER catches decimal, octal, hexadecimal, binary, foating point (scientific noation) and even imaginary number. Too broad for SDL. 
value: function | expr | '-'? NUMBER | NAME | STRING

expr:  (value operator value) | ( '(' value operator value ')' )

#NOTE: & and && not in 14496-1 but used in 14496-12
operator: operator_test | operator_logical | operator_bin | operator_math

operator_math: '+' | '-' | '/' | '*'

operator_test: '==' | '<=' | '<' | '>=' | '>' | '!='

operator_bin: '&' | '|'

operator_logical: '&' '&' | '|' '|'

function: function_name '(' value ')'

#NOTE: Only lengthof in 14496-1, floor is used in 14496-12 without definition
function_name: 'floor' | 'lengthof'

array_initialisation: '{' ','.value+ '}'

# Rule FC.1
if_stmt: 'if' '(' condition ')' '{' body '}' else_if_stmt? else_stmt?

else_if_stmt: 'else' 'if' '(' condition ')' '{' body '}'

else_stmt: 'else' '{' body '}'

condition: value

# Rule FC.2
switch_stmt: 'switch' '(' condition ')' '{' (switch_case switch_break?)* switch_default? switch_break?'}'

switch_break: 'break' ';'
switch_case: 'case' (NUMBER | NAME | STRING) ':' body? 
switch_default: 'default' ':' body?

# Rule FC.3
for_stmt: 'for' '(' expression1 ';' expression2 ';' expression3 ')' '{' body '}'

for_variable_declaration_assignment: type NAME array_length? assigned_value

expression1:
  | variable_assignment
  | for_variable_declaration_assignment


expression2: value

expression3: variable_incr

# Rule FC.4
do_stmt: 'do' '{' body '}' 'while' '(' condition ')' ';'

# Rule FC.5
while_stmt: 'while' '(' condition ')' '{' body '}'

